capture log close
log using suic-econ-a3-1994.txt, replace text

//  program:    suic-econ-a3-1994.do
//  task:		estimate impact of recessions on suicide, restricting to 
//				CPS dataset for years 1994-2010.
// 	input:		suic-econ-data2-cps.dta
//	output:		suic-econ-total1994.txt
//  project:    economic conditions and suicide
//  author:     sam harper \ 17dec2014

//  #0
//  program setup

version 12
set linesize 80
clear all
macro drop _all
set matsize 11000


// #1
// data and descriptive table

* CPS dataset with education, 1994-2010
use suic-econ-data2-cps, clear

* re-scale icei for a 10-unit change, change sign to estimate decrease
replace icei= -1*icei/10



// #2
// crude poisson model without adjustments: Model 0

poisson suic icei, offset(logpop100y) vce(cluster state) nolog ///
	irr cformat(%4.3f)

* predicted marginal effects
eststo m_0: estpost margins, dydx(icei) predict(ir) ///
	vce(unconditional) cformat(%4.2f)

* save estimates for plot
qui parmest, saving(te_1994__m_0, replace) 
	
	
// #3
// now add individual covariates: Model 1

poisson suic icei i.sex i.race2 i.agecat i.educa3, offset(logpop100y) ///
	vce(cluster state) nolog irr cformat(%4.3f)

* predicted margins
eststo m_1: estpost margins, dydx(icei sex race2 agecat educa3) predict(ir) ///
	vce(unconditional) cformat(%4.2f)

* save estimates for plot
qui parmest, saving(te_1994__m_1, replace) 	


// #4
// now add state, season, and year fixed effects: Model 2

poisson suic icei i.sex i.race2 i.agecat i.educa3 i.stabb i.season##i.year, ///
	offset(logpop100y) vce(cluster state) nolog irr cformat(%4.3f)

* predicted marginal effects
eststo m_2: estpost margins, dydx(icei sex race2 agecat educa3 season) ///
	predict(ir) vce(unconditional) cformat(%4.2f) 
	
* save estimates for plot
qui parmest, saving(te_1994__m_2, replace) 


// #5
// now add state-specific linear trends to FE model: Model 3

poisson suic icei i.sex i.race2 i.agecat i.educa3 i.stabb##c.quarter ///
	i.season##i.year, offset(logpop100y) vce(cluster state) nolog ///
	irr cformat(%4.3f)

* predicted marginal effects
eststo m_3: estpost margins, dydx(icei sex race2 agecat educa3 season) ///
	predict(ir) vce(unconditional) cformat(%4.2f) 

* save estimates for plot
qui parmest, saving(te_1994__m_3, replace) 	
	

// #6
// now add state-specific quadratic time trends to basic FE model: Model 4

poisson suic icei i.sex i.race2 i.agecat i.educa3 i.season##i.year ///
	i.stabb##(c.quarter##c.quarter), offset(logpop100y) vce(cluster state) ///
	nolog irr cformat(%4.3f)

* predicted margins
eststo m_4: estpost margins, dydx(icei sex race2 agecat educa3 season) ///
	predict(ir) vce(unconditional) cformat(%4.2f) 

* save estimates for plot
qui parmest, saving(te_1994__m_4, replace) 

	
// #7 	
// include state-specific cubic trends: Model 5

poisson suic icei i.sex i.race2 i.agecat i.educa3 i.season##i.year ///
	i.stabb##(c.quarter##c.quarter##c.quarter), offset(logpop100y) ///
	vce(cluster state) nolog irr cformat(%4.3f)

* predicted margins
eststo m_5: estpost margins, dydx(icei sex race2 agecat educa3 season) ///
	predict(ir) vce(unconditional) cformat(%4.2f) 

* save estimates for plot
qui parmest, saving(te_1994__m_5, replace) 	

// #8
// write results to file

esttab m_* using suic-econ-total1994.txt,  ///
	cells("b(fmt(%4.2f)) ci(par( ( , ) ))") collabels(Est. 95%CI) nostar ///
	wide not nolines nonum mtitle((0) (1) (2) (3) (4) (5)) ///
	delimiter(";") replace
	

log close
exit	
